A potential disintegration of the West Antarctic ice sheet: Implications for economic analyses of climate policy

Delavane Diaz and Klaus Keller[†]

Online Appendix

1 Supplemental Results



Figure SM 1: Reliability of avoiding triggering WAIS disintegration by 2100 for optimal DICE-WAIS Monte Carlo runs with parametric uncertainty about climate sensitivity, WAIS trigger temperature, and rate of WAIS discharge. Color gradient of scenario map classifies parametric uncertainty from low threat (e.g., blue in upper left shows low climate sensitivity and high trigger temperature) to high threat (e.g., red and black in lower right shows high climate sensitivity and low trigger temperature). Black shows a reliability of 0%.

[†]Diaz: Electric Power Research Institute, 1325 G Street NW Suite 1080, Washington, DC 20005 (e-mail: ddiaz@epri.com); Keller: Penn State University, 436 Deike Building, University Park, PA 16802 (e-mail: klaus@psu.edu).



Figure SM 2: 2020 optimal Social Cost of Carbon (\$/ton CO_2) for optimal DICE-WAIS Monte Carlo runs.



Figure SM 3: 2020 optimal Social Cost of Carbon (\$/ton CO_2) for sensitivity cases.



Figure SM 4: WAIS reliability by mitigation strategy. Point cloud illustrates WAIS reliability as a function of mitigation strategy with parametric uncertainty about climate sensitivity, WAIS trigger temperature, and rate of WAIS discharge. Low, medium, and high threat scenarios are classified on the scenario map in Figure SM 1.



Figure SM 5: WAIS reliability as a function of carbon tax policy. Points reflect relationship between CO_2 price and reliability outcome for WAIS for Monte Carlo experiment. Low, medium, and high threat scenarios are classified on the scenario map in Figure SM 1.



Figure SM 6: Distribution of global mean sea level rise in 2100 (m).



Figure SM 7: Distribution of WAIS melt rate in 2100 (mm/yr).

2 DICE-WAIS Model Documentation

DICE-WAIS is a stochastic programming integrated assessment model (IAM) with endogenous uncertainty about the possible disintegration of the West Antarctic ice sheet (WAIS), based on the Dynamic Integrated Climate-Economy (DICE) model version 2013R (Nordhaus and Sztorc, 2013). This analysis uses DICE as a benchmark IAM for simplicity and speed, but the stochastic catastrophe framework described here could be flexibly integrated with many other IAMs.

2.1 DICE Overview

DICE is a transparent and tractable intertemporal optimization model of economic growth and climate impacts for a single region, the world. DICE solves the optimal Pareto problem, which sets the level of greenhouse gas mitigation such that marginal cost of mitigation is equal to the marginal benefit of avoided climate impacts over the model time path. DICE chooses the optimal path of consumption that maximizes the social welfare objective function.[1] A stylized representation of DICE is shown in Figure SM 8.



Figure SM 8: Stylized diagram of the DICE integrated assessment model.

Welfare is the discounted sum of utility over time, where the isoelastic (i.e., constant relative risk aversion) utility function expresses preferences over per capita consumption:

$$W = \sum_{t \in T} \frac{1}{(1+\rho)^t} \left[l_t \frac{(\frac{C_t}{l_t})^{(1-\eta)} - 1}{1-\eta} \right] \tag{1}$$

where W is total social welfare, C_t is the level of consumption, l_t is the population, ρ, the pure rate of social time preference, and η is the consumption elasticity parameter. Utility, the second term

[1] DICE is a neoclassical model of optimal economic growth premised on the Ramsey rule $r_t = \rho + \eta g_t$, where r_t is the discount rate, ρ is the pure rate of time preference, η is the marginal utility of consumption, and g_t is the per capita growth rate of consumption (Ramsey 1928).

in the formula, is weighted by the social discount factor, the first term. The parameter ρ reflects intertemporal preferences for comparing utility across different generations.

In DICE utility increases in population and per capita consumption, with diminishing marginal utility from the latter. η, the elasticity of marginal utility of consumption, captures aversion to inequality in per capita consumption levels. These two preference parameters are calibrated in DICE in accordance with the Ramsey growth equation to observed economic outcomes (e.g., interest rates and rates of return on capital) such that ρ is 1.5 and η is 1.45 (Nordhaus and Sztorc, 2013).

Economic output is determined by a Cobb-Douglas production function of endogenous capital and exogenous labor, with exogenous Hicks-neutral technological change represented by total factor productivity. This output, if unmitigated, has an associated carbon intensity, resulting in greenhouse gas emissions that warm the atmosphere. The climate module computes CO_2 concentrations, radiative forcing, and atmosphere and ocean warming.

The decision variable for carbon mitigation, μ, equals the fraction of emissions from the business-as-usual emissions projection that are avoided through decarbonization. The cost of mitigation (as a proportion of output) is given by a convex power function of μ in which the marginal cost of mitigation increases more than linearly with μ. Climate damages Ω act as a claim on output, reducing the amount that can be spent on either welfare-improving consumption today or investment in the future capital stock. The DICE-2013R model documentation and GAMS code are described in Nordhaus and Sztorc (2013).

2.2 DICE-WAIS Stochastic Optimization with Endogenous Uncertainty

We modify DICE to investigate the implications of a possible disintegration of WAIS. The modified model, DICE-WAIS, is formulated as a multistage stochastic programming framework, based on the approach of Rutherford (2013). This innovative solution method uses a sequential binomial scenario tree to parsimoniously describe the catastrophic 'states of the world' (Figure SM 9), allowing the optimization problem to be formulated as the deterministic equivalent and efficiently solved.[2]

[2] The stochastic programming framework is conceptually similar to the more complex structure of the stochastic dynamic programs employed in Cai, Judd and Lontzek (2013) and Lemoine and Traeger (2014), but avoids the computational burden of backward recursion and value function approximation, reducing solve times by more than an order of magnitude. Rutherford classifies this approach as stochastic control, to distinguish the fact that the hazard rate probabilities are endogenous to the model, compared to a traditional stochastic program with exogenous probabilities for each state.



Figure SM 9: Scenario tree for stochastic programming catastrophe framework. Modified from Rutherford (2013). The hazard rate hr_t gives the probability of catastrophe in period t conditional on there having been no catastrophe up to period t. This example uses 10 year time periods but the framework can be flexibly defined.

A catastrophe can be triggered in any time period t, where the hazard rate hr_t represents the probability of a catastrophe occurring based on warming at time t, conditional on the fact that one has not yet occurred. At the outset, the likelihood is uncertain, but the decision-maker is assumed to have prior knowledge about how the hazard rate relates to temperature as well as the consequence of the event. By integrating this stochastic framework with the IAM, the decision-maker can influence current and future probabilities of catastrophe (i.e., the time series of hazard rates) by mitigating CO_2 to reduce climate warming.

We integrate Rutherford's general framework into DICE, and also introduce additional variables related to sea level rise (SLR) and the associated economic damages from coastal impacts. The DICE-WAIS model is solved along each one of the disintegration branches of the stochastic scenario tree presented in Figure SM 9. The key difference among the different scenario branches is a shift in the SLR regime. The DICE-WAIS diagram in Figure SM 10 depicts the blue 'pre-trigger' pathway that corresponds to the upper branch of the scenario tree, while the red 'post-trigger' pathway corresponds to the offshoot disintegration scenarios.

SLR dynamics in the 'no catastrophe' pre-trigger state of the world are based on DICE-2010, the first and only vintage of the DICE model to explicitly include an SLR module (Nordhaus, 2010*b*). DICE-2010 decomposes SLR into contributions from four major processes: thermal expansion, melt from glaciers and small ice caps, Greenland Ice Sheet melt, and Antarctic Ice Sheet melt, parameterized in accordance with the IPCC Fourth Assessment Report (AR4). Thermal expansion reaches a steady state equilibrium of 0.5m per 1 °C of warming by year 3000 at a rate of 2% per decade based on Earth System Models of Intermediate Complexity. Equilibrium is reached . Glaciers and small ice caps are a minor contributor to SLR, limited to 0.26m of SLR-equivalent at a melt rate of 0.0008m per year per °C. Greenland Ice Sheet melt is calibrated to Ridley et al. (2005), with a melt rate of 0.11mm per year per °C.

We use these first three components to describe a 'baseline SLR' function of temperature:

$$SLR_{t+1} = (r_{TE} + r_{GSIC} + r_{GIS})\,\Delta T_t + SLR_t \tag{2}$$



Figure SM 10: Stylized diagram of the DICE-WAIS model.

The fourth component, SLR from WAIS melt, is only introduced in the post-trigger catastrophic states of the world, as described in Section SM 2.2.2.

There are two types of climate damages in DICE-WAIS, temperature damages and SLR damages. Both damage functions are based on the DICE-2010 model, which distinguishes two categories of impact sectors, coastal impacts and all other noncoastal impacts (Nordhaus, 2010*a*). Temperature damages are a function of ΔT as given in Equation SM 3. These damages include all market and nonmarket impacts excluding coastal impacts. For these damages, Ω_{TEMPt} gives the fraction of economic output lost to temperature during time period t, formulated as a quadratic function of ΔT_t, the equilibrium change in global mean surface temperature above preindustrial. The severity of the damage function is specified by the linear coefficient a_0 and quadratic coefficient a_1:

$$\Omega_{TEMPt} = \alpha_1 \Delta T_t + \alpha_2 \Delta {T_t}^2 = 0.00008 \Delta T_t + 0.002 \Delta {T_t}^2 \tag{3}$$

The other type of damages is SLR damages. Ω_{SLRt} gives the fraction of economic output lost to coastal damages during time period t, a function of SLR_t, without regard to the source of the SLR (e.g., baseline or from WAIS).

$$\Omega_{SLR_t} = \beta_1 SLR_t + \beta_2 {SLR_t}^2 = 0.00518 SLR_t + 0.00306 {SLR_t}^2 \tag{4}$$

Finally, we reformulate the model in 10 year time steps in order to extend the computational time horizon. This involves adjusting parameters of the carbon cycle and other period rates. The following subsections will describe the characterization of both the hazard and the consequence of WAIS disintegration.

2.2.1 Hazard of WAIS Disintegration

We use a hazard rate approach to approximate the complex geophysics of WAIS disintegration. Hazard rates are a tractable way to model the stochastic nature of an uncertain climate catastrophe and are frequently used in survival analysis to represent the likelihood of an event at time t, conditional on survival until that time. In order to represent the expected relationship between likelihood and warming, we assume a stylized functional form for the hazard rate. Specifically, the probability of triggering a disintegration is defined to be a quadratic function of global mean temperature change

$$hr_t = \min[\beta\,(\Delta T_t)^2\,, 1] \tag{5}$$

where hr_t is the hazard rate for period t, the probability of catastrophe in period t conditional on there having been no catastrophe up to period t, β is the disintegration coefficient, and ΔT_t is global mean temperature change.

Calibrating the hazard rate function (Equation SM 5) requires some subjective belief about the probability of WAIS disintegration for a given climate scenario, as specialized ice sheet models face obstacles such as computational limitations to resolve ice flow dynamics (e.g., the migrating calving front or grounding line) in three-dimensional space and thus are currently thought to be incomplete. Three surveys of expert assessments characterizing the hazard of WAIS disintegration are especially relevant to our study, each surveying a different aspect of the problem. Most recently, Bamber and Aspinall (2013) elicited expert opinion about the contribution of the three polar ice sheet components to global SLR in 2100 and found a mean melt rate for WAIS of 3 mm/yr, with the 95th percentile melt rate of 11.8 mm/yr. Kriegler et al. (2009) conducted an expert elicitation on the likelihood of five potential climate tipping points occurring by 2200. For WAIS disintegration they found a mid-range probability of 0.6 from core experts assuming a high temperature scenario (3-5.5°C above pre-industrial in 2100). Finally, a risk estimation study by Vaughan and Spouge (2002) assessed an expert panel about the probability of WAIS disintegration by 2200. The result of this study was a probability of 0.05 that the melt rate of WAIS would be 10 mm/yr, with a probability of 0.3 that it would be 2 mm/yr. Overall there is a trend of the likelihoods increasing over time. Moreover, two recent observational and modeling studies suggest that a WAIS disintegration is nearly certain at some point in the future, assuming a business-as-usual emissions pathway (Joughin, Smith and Medley, 2014; Rignot et al., 2014).

Specifically, we interpret and average expert opinion about the likelihood of WAIS disintegration under alternative warming scenarios using the results of Kriegler et al. (2009).



Figure SM 11: Interpreted distribution based on Kriegler et al. (2009) expert opinion about temperature pathway triggering WAIS collapse by 2200.

To calibrate the hazard function of temperature change in Equation SM 5, We solve a system of equations for β, the desired hazard rate calibration coefficient:

$$P_t = hr_t \Pi_t \tag{6}$$

$$hr_t = \min[\beta\,(\Delta T_t)^2\,, 1] \tag{7}$$

$$\Pi_t = \prod_{i<t}(1 - hr_i) \tag{8}$$

$$\sum_{t \leq 2200} P_t = 1 \tag{9}$$

where state s and time period t are defined in 10-year steps between 2010 and 2100, P_s is the probability of being in state s, hr_t is the disintegration hazard rate, Π_t is the probability that the disintegration has not yet occurred at start of time t, and ΔT_t is the global mean temperature change given by the warming timepath.

2.2.2 Consequence of WAIS Collapse

The consequence of triggering the WAIS disintegration is represented in the stochastic framework as a regime shift in the climate system, to the 'post-trigger' pathway in the DICE-WAIS diagram (Figure SM 10). We use disintegration to mean the onset of additional SLR from WAIS and the start of an irreversible process.[3] This irreversible design intends to reflect current understanding of ice sheet dynamics (e.g., ice sheet growth occurs over tens of thousands of years through accumulated snowfall, and will not be easily re-formed; dynamic instabilities suggest it may not be possible to

[3] These events may not coincide in time. For instance, the "point of no return" may be crossed before rapid melt begins; similarly, rapid melt may begin, but there is still a window of opportunity to prevent the total disintegration of the sheet. Moreover, despite unmanned monitoring rovers and tide gauge network, it is not guaranteed that the onset of rapid melt will be immediately detected in the observational record.

halt the process of ice flow once underway).

The SLR contribution of WAIS in the disintegration state of the world is assumed to be a constant annual rate. This rate is modeled as an uncertain parameter based on the Bamber and Aspinall (2013) survey of WAIS contributions to global SLR rates in 2100. Specifically, we assume that the annual rate of disintegration is lognormally distributed with a mean of 3.3 mm/yr and standard deviation 1.65 mm/yr. This implies that the average timeframe of complete disintegration of WAIS is 1,000 years, which is consistent with expert estimates and covers the literature ranging from 400 to 2400 years (National Research Council, 2013; Oppenheimer, 1998). As a check on this assumption, we compare our realized annual rate of SLR from WAIS in 2100 (Figure SM 7) to the published distribution in Bamber and Aspinall (2013).

The direct physical consequence of WAIS disintegration is the additional rate of SLR described above.[4] In an IAM framework, this physical consequence of SLR has important implications in terms of social welfare. These costs to society are represented through the coastal damage function of SLR (Equation SM 4).

2.3 Study Methods

In this study we apply the DICE-WAIS model described above to evaluate the Pareto optimal mitigation strategy that balances the uncertain climate damage outcomes across all possible states of the world with the costs of mitigation investments (see e.g., Diaz, 2015). In addition to the overarching stochastic uncertainty about the WAIS threat, this analysis considers parametric uncertainty in a Monte Carlo experiment with 5,000 model runs using Latin Hypercube Sampling from probability distributions for three key uncertain parameters: climate sensitivity (from Olson et al., 2012), trigger temperature (from Kriegler et al., 2009), and the annual rate of WAIS discharge (from Bamber and Aspinall, 2013). We also conduct sensitivity analysis with different assumptions about the slope of the SLR damage function and the social rate of time preference.

[4] The physical consequence of WAIS disintegration is limited to additional SLR. This constrained scope does not fully capture the effect of WAIS disintegration on climate system dynamics, which could alter the local weather patterns as well as the albedo (surface reflectivity). In future work it would be interesting to explore this additional dimension of the climate system.

A DICE-WAIS Model Code

DICE-WAIS is programmed in GAMS (Brooke, Kendrick and Meeraus, 1988) and solved with the CONOPT nonlinear solver (Drud, 1996). Data and code are publicly available at the web repository `https://github.com/delavane/DICEWAIS`.

```
*This is edited from the DICE-2013R model, version DICE2013Rv2_102213_vanilla_v24b.gms
* Delavane Diaz, ddiaz@epri.com
* Last revised Jan 8, 2016
$title Modeling the threat of WAIS collapse with endogenous uncertainty
$eolcom #

* DICE2013, modified to run in 10 year time steps (vs 5 yr) from 2010
$if not set yr $set yr 30 # Set total time horizon
sets sequence /0/
ct total climate time periods /1*%yr%/
t(ct) time periods /1*%yr%/
collapset collapse time periods /1*15/
tfirst(t) first time period
;
alias(t,tt);
tfirst(t) = yes$(t.val eq 1);

parameters
tstep Years per Period /10/
** Preferences
elasmu Elasticity of marginal utility of consumption (1.45 default computed below)
prstp Initial rate of social time preference per year /%prtp%/

** Population and technology
gama Capital elasticity in production function/.300/
pop0 Initial world population (millions) /6838/
popadj Growth rate to calibrate to 2050 pop projection /0.134/
popasym Asymptotic population (millions) /10500/
dk Depreciation rate on capital (per year) /.100/
q0 Initial world gross output (trill 2005 USD) /63.69/
k0 Initial capital value (trill 2005 USD) /135/
a0 Initial level of total factor productivity /3.80/
ga0 Initial growth rate for TFP per period (0.079 per 5 yr --> 1.079^2-1=0.164241) /0.164/
dela Decline rate of TFP per year /0.006/
l(ct) Level of population and labor
al(t) Level of total factor productivity
sigma(t) CO2-equivalent-emissions output ratio
rr(ct) Average utility social discount rate
ga(t) Growth rate of productivity from
gl(t) Growth rate of labor
*gcost1 Growth of cost factor (not used)
gfacpop(t) Growth factor population

** Emissions parameters
gsigma1 Initial growth of sigma (per year) /-0.01/
dsig Decline rate of decarbonization (per YEAR - typo in code?  period) /-0.001/
eland0 Carbon emissions from land 2010 (GtCO2 per year) /3.3/
deland Decline rate of land emissions (per period) /.2/
e0 Industrial emissions 2010 (GtCO2 per year) /33.61/
miu0 Initial emissions control rate for base case 2010 /0 / # DICE2013 uses.039
forcoth(t) Exogenous forcing for other greenhouse gases
etree(t) Emissions from deforestation

** Carbon cycle
* Initial Conditions
mat0 Initial Concentration in atmosphere 2010 (GtC)/830.4/
mu0 Initial Concentration in upper strata 2010 (GtC) /1527./
ml0 Initial Concentration in lower strata 2010 (GtC) /10010./
mateq Equilibrium concentration atmosphere (GtC) /588/
mueq Equilibrium concentration in upper strata (GtC) /1350/
mleq Equilibrium concentration in lower strata (GtC) /10000/
* Flow paramaters
* Use different parameters to reflect the 10 year time step -- percent per period
b12 Carbon cycle transition matrix - fraction of conc that goes into upper ocean (0.088 per 5 yr) /0.176/
b23 Carbon cycle transition matrix (0.00250 per 5 yr) /0.005/
* These are for declaration and are defined later
```

```
b11 Carbon cycle transition matrix
b21 Carbon cycle transition matrix
b22 Carbon cycle transition matrix
b32 Carbon cycle transition matrix
b33 Carbon cycle transition matrix
sig0 Carbon intensity 2010 (kgCO2 per output 2005 USD 2010)

** Climate model parameters
*Use different parameters to reflect the 10 year time step
t2xco2 Equilibrium temp impact (oC per doubling CO2) /2.9/
fex0 2010 forcings of non-CO2 GHG (Wm-2) /0.25/
fex1 2100 forcings of non-CO2 GHG (Wm-2) /0.70/
tocean0 Initial lower stratum temp change (C from 1900) /.0068/
tatm0 Initial atmospheric temp change (C from 1900) /0.80/
c10 Initial climate equation coefficient for upper level (0.098 per 5 yr --> keep this as 5 yr and correct below)
/0.098/
c1beta Regression slope coefficient(SoA~Equil TSC) /0.01243/
c1 Climate equation coefficient for upper level (0.098 per 5 yr) /0.196/ # SoA Speed of adjustment
c3 Transfer coefficient upper to lower stratum (0.088 in DICE2013 - time independent?)  /0.088/
c4 Transfer coefficient for lower level (0.0250 per 5 yr) /0.05/
fco22x Forcings of equilibrium CO2 doubling (Wm-2) /3.8/
slr0 Initial sea level rise since 2000 (m) in 2010 per Church and White 2011 /0.04/
lam Climate model parameter
waisrate WAIS collapserate (m per year)
beta Decadal hazard rate coefficient as a Markovian function
WAISthresholdT WAIS threshold

** Climate damage parameters per DICE 2010 which splits SLR and Temp damages
a1 Damage coefficient on temperature /0.00008162/
a2 Damage coefficient on temperature squared/0.00204626/
a3 Exponent on temperature damages /2/
b1 Damage coefficient on SLR/0.00518162/
b2 Damage coefficient on SLR squared/0.00305776/
b3 Exponent on SLR damages /2/

** Abatement cost
expcost2 Exponent of control cost function /2.8/
pback Cost of backstop 2005$ per tCO2 2010 /344/
gback Initial cost decline backstop cost per period /.025/
limmiu(t) Upper limit on control rate (allow 20% negative emissions after 2100)
tnopol Period before which no emissions controls base /45/
cprice0 Initial base carbon price (2005$ per tCO2) /1.0/
gcprice Growth rate of base carbon price per year /.02/
gsig(t) Change in sigma (cumulative improvement of energy efficiency)
cost1(t) Adjusted cost for backstop
pbacktime(t) Backstop price
optlrsav Optimal long-run savings rate used for transversality
cpricebase(t) Carbon price in base case
** Availability of fossil fuels
fosslim Maximum cumulative extraction fossil fuels (GtC) /6000/
;

* Define all uncertain / sensitivity parameters
* Read in uncertain parameters and make any adjustments by overwriting
t2xco2 = %ECS%;
waisrate = %WAISrate%/1000; # Convert WAIS collapserate from mm/yr to m/yr (e.g., 0.01 m per year Bamber 2013)
beta = 0.6255*%Ttrigger%**(-1.932);

$if not set montecarlo $goto continue
* Read in uncertain parameters and make any adjustments
parameters inputdata, reltarget, tax;
$gdxin DICEWAISinputs
$load inputdata
t2xco2 = inputdata('%iteration%','ECS');
waisrate = inputdata('%iteration%','WAISrate')/1000; # Convert WAIS collapserate from mm/yr to m/yr (e.g., 0.01
m per year Bamber 2013)
beta = inputdata('%iteration%','hazardbeta');
tax = %tax%;

$label continue
*Transient TSC Correction ("Speed of Adjustment Parameter")
c1 = 2*(c10 + c1beta*(t2xco2-2.9)); # 2x --> to reflect 10 yr periods
* Maintain same linear and quadratic ratio in SLR DF
b1=%slrDF%*0.00518162/0.00823938;
b2=%slrDF%*0.00305776/0.00823938;
* adjust consumption elasticity when we alter prstp:  r=elasmu*g+prstp with r=0.04623, g=0.02154
elasmu = (0.04623-%prtp%)/0.02154;

$if %scenario% == tax a1=0; a2=0; b1=0; b2=0;

* Parameters for long-run consistency of carbon cycle
b11 = 1 - b12;
b21 = b12*MATEQ/MUEQ;
b22 = 1 - b21 - b23;
```

```
b32 = b23*mueq/mleq;
b33 = 1 - b32 ;
* Further definitions of parameters
sig0 = e0/(q0*(1-miu0));
lam = fco22x/t2xco2;
l("1") = pop0;
loop(t, l(t+1)=l(t););
loop(t, l(t+1)=l(t)*(popasym/L(t))**popadj ;);
loop(ct$(not t(ct)), l(ct)=l(ct-1); ); # hold labor constant after end of time horizon
ga(t)=ga0*exp(-dela*tstep*((t.val-1)));
al("1") = a0; loop(t, al(t+1)=al(t)/((1-ga(t)));) ;
gsig("1")=gsigma1; loop(t,gsig(t+1)=gsig(t)*((1+dsig)**tstep) ;);
sigma("1")=sig0; loop(t,sigma(t+1)=(sigma(t)*exp(gsig(t)*tstep));) ;
pbacktime(t)=pback*(1-gback)**(t.val-1);
cost1(t) = pbacktime(t)*sigma(t)/expcost2/1000;
etree(t) = eland0*(1-deland)**(t.val-1);
rr(ct) = 1/((1+prstp)**(tstep*(ct.val-1)));
forcoth(t) = fex0+ (1/18)*(fex1-fex0)*(t.val-1)$(t.val lt 19)+ (fex1-fex0)$(t.val ge 19);
optlrsav = (dk + .004)/(dk + .004*elasmu + prstp)*gama;
*Base Case Carbon Price
cpricebase(t)= cprice0*(1+gcprice)**(tstep*(t.val-1));
limmiu(t)=1; # except cases below:
limmiu('1')=0.1;
limmiu(t)$(t.val>11)=1.2;


** This section corresponds to the model case specified earlier (e.g., neglect_collapse, collapse, collapse%certain%,
ev_collapse
$goto %model%

$label neglect_collapse
set s states (year of collapse) /0/
as(s,t) active states;
parameter
o(s,t) state offset pointer
collapse(s,t) indicator for collapse trigger (0 or 1)
;
collapse(s,t) = 0;
as(s,t) = yes;
o(s,t) = 0;
$goto model

$label collapse
set
s states (year of collapse) /set.sequence,set.collapset/ # collapse horizon can be less than the total horizon
as(s,t) active states;
parameter
o(s,t) state offset pointer
collapse(s,t) indicator for collapse trigger (0 or 1)
$if set SP pr(s) state probability # note probabilities will be fixed exogenously
;
collapse(s,t) = 0;
loop(s$(not sameas(s,'0')),
o(s,t)$(t.val<=s.val) = -s.val;
as(s,t) = yes$(o(s,t) = 0);
* If a collapse occurs, increased slr rate does not apply until start of NEXT time period
collapse(s,t)$(t.val > s.val) = 1;
);
o('0',t) = 0;
as('0',t) = yes;
$goto model


** Main model code continues here
$label model
$macro sw(s,t) s+o(s,t),t

VARIABLES
MIU(s,t) Emission control rate GHGs
FORC(s,t) Increase in radiative forcing (watts per m2 from 1900)
TATM(s,t) Increase temperature of atmosphere (degrees C from 1900)
TOCEAN(s,t) Increase temperatureof lower oceans (degrees C from 1900)
MAT(s,t) Carbon concentration increase in atmosphere (GtC from 1750)
MU(s,t) Carbon concentration increase in shallow oceans (GtC from 1750)
ML(s,t) Carbon concentration increase in lower oceans (GtC from 1750)
E(s,t) Total CO2 emissions (GtCO2 per year)
EIND(s,t) Industrial emissions (GtCO2 per year)
C(s,t) Consumption (trillions 2005 US dollars per year)
K(s,t) Capital stock (trillions 2005 US dollars)
CPC(s,t) Per capita consumption (thousands 2005 USD per year)
I(s,t) Investment (trillions 2005 USD per year)
SR(s,t) Gross savings rate as fraction of gross world product
RI(s,t) Real interest rate (per annum)
Y(s,t) Gross world product net of abatement and damages (trillions 2005 USD per year)
```

```
YGROSS(s,t) Gross world product GROSS of abatement and damages (trillions 2005 USD per year)
DAMAGES(s,t) Damages (trillions 2005 USD per year)
DAMFRAC(s,t) Damages as fraction of gross output
ABATECOST(s,t) Cost of emissions reductions (trillions 2005 USD per year)
MCABATE(s,t) Marginal cost of abatement (2005$ per ton CO2)
CCA(s,t) Cumulative industrial carbon emissions (GTC)
PERIODU(s,t) One period utility function
CPRICE(s,t) Carbon price (2005$ per ton of CO2)
CEMUTOTPER(s,t) Period utility
UTILITY(s) Welfare function
EU Expected Utility
* Define states and probabilities
PR(s) state probability
HR(t) collapse hazard rate
PI(t) probability of no collapse at start of time t
* Coastal impacts
WAIS(s,t) SLR component from WAIS collapse in terms of m per period
SLR(s,t) total sea level rise in year t under state scenario s (m)
COASTDAMAGES(s,t) Coastal damages ($T)
;
NONNEGATIVE VARIABLES MIU, TATM, MAT, MU, ML, Y, YGROSS, C, K, I, SLR, DAMAGES, COASTDAMAGES, PR, HR, PI;

EQUATIONS
*Emissions and Damages
EEQ(s,t) Emissions equation
EINDEQ(s,t) Industrial emissions
CCACCA(s,t) Cumulative carbon emissions
FORCE(s,t) Radiative forcing equation
DAMFRACEQ(s,t) Equation for damage fraction
DAMEQ(s,t) Damage equation
ABATEEQ(s,t) Cost of emissions reductions equation
MCABATEEQ(s,t) Equation for MC abatement
CARBPRICEEQ(s,t) Carbon price equation from abatement
*Climate and carbon cycle
MMAT(s,t) Atmospheric concentration equation
MMU(s,t) Shallow ocean concentration
MML(s,t) Lower ocean concentration
TATMEQ(s,t) Temperature-climate equation for atmosphere
TOCEANEQ(s,t) Temperature-climate equation for lower oceans
*Economic variables
YGROSSEQ(s,t) Output gross equation
YY(s,t) Output net equation
CC(s,t) Consumption equation
CPCE(s,t) Per capita consumption definition
SEQ(s,t) Savings rate equation
KK(s,t) Capital balance equation
RIEQ(s,t) Interest rate equation
* Utility
CEMUTOTPEREQ(s,t) Period utility
PERIODUEQ(s,t) Instantaneous utility function equation
UTIL(s) Objective function
WELFARE Objective function
* Define states and probabilities
HRDEF(t) Defines HR
PIDEF(t) Defines PI
PRDEF(s,t) Defines PR
PR0(s) Ensure no collapse scenario accounted for
* Coastal impacts
WAISDEF(s,t) Define SLR component from WAIS collapse in terms of m per period
SLRDEF(s,t) Define total sea level rise in year t under state s (m)
COASTDAMAGESDEF(s,t) Define coastal damages ($T)
;

** Characterize WAIS collapse hazard
set tp(t) Time periods with tipping points;
tp(t) = yes$(t.val <= smax(s,s.val));
$if %model%==collapse%certain% tp(t) = yes$(t.val = smax(s,s.val));

* Hazard rate depends on the temperature in s=0 world
* beta has been calibrated as a decadal hazard rate of warming since 2000
HRDEF(tp(t))$(t.val>1)..  HR(t) =e= beta*(TATM('0',t)-0.6)**2;

* Hazard rate parameters beta and alpha are calibrated offline in DICEWAIS_uncertain_parameters.R
HR.FX('1')=0;

* Probability that collapse has not occurred at start of time period t (this is only enforced for periods that have
tipping potential)
PIDEF(tp(t))..  PI(t) =e= prod(tt$(ord(tt) lt ord(t)), 1-HR(tt) );

* Probability that a tipping point occurs in time period t:
PRDEF(s,t)$sameas(s,t)..  PR(s) =e= HR(t) * PI(t);
PR0(s)$(s.val=0)..  PR(s) =e= 1 - sum(tp(t), HR(t) * PI(t));
```

```
** Equations of the DICE 2013 model
$label DICE
*Emissions and Damages
eeq(as(s,t))..  E(s,t) =E= EIND(s,t) + etree(t); # annual emissions
eindeq(as(s,t))..  EIND(s,t)=E= sigma(t) * YGROSS(s,t) * (1-(MIU(s,t)));
ccacca(as(s,t+1))..CCA(s,t+1) =E= CCA(sw(s,t))+ EIND(sw(s,t))*tstep/3.666;
force(as(s,t))..  FORC(s,t)=E= fco22x * ((log((MAT(s,t)/588.000))/log(2))) + forcoth(t);
damfraceq(as(s,t))..  DAMFRAC(s,t) =E= a1*TATM(s,t) + a2*TATM(s,t)**a3;
dameq(as(s,t))..  DAMAGES(s,t) =E= YGROSS(s,t) * DAMFRAC(s,t);
abateeq(as(s,t))..  ABATECOST(s,t) =E= YGROSS(s,t) * cost1(t) * (MIU(s,t)**expcost2);
mcabateeq(as(s,t))..  MCABATE(s,t) =E= pbacktime(t) * MIU(s,t)**(expcost2-1);
carbpriceeq(as(s,t))..  CPRICE(s,t) =E= pbacktime(t) * MIU(s,t)**(expcost2-1);

*Climate and carbon cycle
*All of these equations now use different parameters to reflect the 10 year time step
mmat(as(s,t+1))..  MAT(s,t+1) =E= MAT(sw(s,t))*b11 + MU(sw(s,t))*b21 + (E(sw(s,t))*(tstep/3.666));
mml(as(s,t+1))..  ML(s,t+1)=E= ML(sw(s,t))*b33 + MU(sw(s,t))*b23;
mmu(as(s,t+1))..  MU(s,t+1)=E= MAT(sw(s,t))*b12 + MU(sw(s,t))*b22 + ML(sw(s,t))*b32;
tatmeq(as(s,t+1))..TATM(s,t+1) =E= TATM(sw(s,t)) + c1 * ((FORC(s,t+1)-(fco22x/t2xco2)*TATM(sw(s,t)))-(c3*(TATM(sw(s,t))-TOCEAN(s
toceaneq(as(s,t+1))..  TOCEAN(s,t+1) =E= TOCEAN(sw(s,t)) + c4*(TATM(sw(s,t))-TOCEAN(sw(s,t)));

*Economic variables
ygrosseq(as(s,t))..YGROSS(s,t) =E= (al(t)*(l(t)/1000)**(1-gama))*(K(s,t)**gama);
yy(as(s,t))..  Y(s,t) =E= YGROSS(s,t) - ABATECOST(s,t) - DAMAGES(s,t) - COASTDAMAGES(s,t) - COASTAD(s,t);
cc(as(s,t))..  C(s,t) =E= Y(s,t) - I(s,t);
cpce(as(s,t))..  CPC(s,t) =E= 1000 * C(s,t) / l(t);
seq(as(s,t))..  I(s,t) =E= SR(s,t) * Y(s,t);
kk(as(s,t+1))..  K(s,t+1) =L= (1-dk)**tstep * K(sw(s,t)) + tstep * I(sw(s,t));
rieq(as(s,t+1))..  RI(s,t) =E= (1+prstp) * (CPC(s,t+1)/CPC(sw(s,t)))**(elasmu/tstep) - 1;

*Utility
periodueq(as(s,t))..  PERIODU(s,t) =E= ((C(s,t)*1000/l(t))**(1-elasmu)-1)/(1-elasmu);
cemutotpereq(as(s,t))..  CEMUTOTPER(s,t) =E= PERIODU(s,t) * l(t) * rr(t);
util(s)..  UTILITY(s)=E= tstep * [sum(t, CEMUTOTPER(sw(s,t)))];
WELFARE..  EU =e= 1e-3*sum(s, PR(s) * UTILITY(s) );

** Additional equations related to SLR
* Compute SLR component from WAIS discharge (m per period)
WAISDEF(as(s,t+1))..  WAIS(s,t) =E=
$if set montecarlo inputdata('%iteration%','WAISbaserate')/1000*tstep +
$if not set montecarlo 0.2833/1000*tstep + # mm/yr per Sheperd et al 2012
collapse(sw(s,t))* # this turns collapse on/off
waisrate*tstep # m/period, constant discharge
;

* "Baseline" SLR per DICE2010 - thermal expansion, GSIC, GIS component (not WAIS) in decadal rates as function of
temp; add WAIS
SLRDEF(as(s,t+1))..  SLR(s,t+1) =E= (0.00779+0.0314*0.26+0.00176*7.3)*TATM(sw(s,t)) + SLR(sw(s,t)) + WAIS(sw(s,t));
# rate terms in m/period

** SLR Damages
COASTDAMAGESDEF(as(s,t))..  COASTDAMAGES(s,t) =E= YGROSS(s,t)*(1-1/(1+b1*SLR(s,t)+b2*SLR(s,t)**b3));

*Limits
CCA.up(s,t) = fosslim;
MIU.up(s,t) = limmiu(t);
SR.up(s,t) = 1;
** Upper and lower bounds for stability
K.LO(s,t) = 1;
MAT.LO(s,t) = 10;
MU.LO(s,t)= 100;
ML.LO(s,t)= 1000;
C.LO(s,t) = 2;
TOCEAN.UP(s,t) = 20;
TOCEAN.LO(s,t) = -1;
TATM.UP(s,t) = 10;
TATM.LO(s,t) = 0.6;
CPC.LO(s,t) = .01;
* Initial conditions
CCA.FX(s,tfirst) = 90;
K.FX(s,tfirst) = k0;
MAT.FX(s,tfirst) = mat0;
MU.FX(s,tfirst) = mu0;
ML.FX(s,tfirst) = ml0;
TATM.FX(s,tfirst) = tatm0;
TOCEAN.FX(s,tfirst) = tocean0;
* DICE-WAIS additions
PI.FX(tfirst) = 1; PI.FX('2') = 1;
PR.L(s)$(ord(s)>1) = 0.1;
MIU.L(s,t) = min(t.val/10,1);
TATM.L(s,t) = tatm0+t.val/10;
MIU.FX(as(s,tfirst)) = 0; # No mitigation has happened in first period
SLR.FX(s,tfirst) = slr0;
CEMUTOTPER.L(as(s,t))=1000;
SR.FX(as(s,t))$(ord(t)>1) = optlrsav;
```

```
** Solution options
option iterlim = 99900;
option reslim = 99999;
option solprint = off;
option limrow = 0;
option limcol = 0;
model DICESLR /all/;

$if %scenario% == BAU MIU.fx(s,t)=0;
$if %scenario% == 2deg TATM.up(s,t) = 2;
$if %scenario% == maxeffort MIU.fx(s,t)=limmiu(t);
$if %scenario% == tax CCA.up(s,t) = inf; MIU.l(s,t)=1; cprice.fx(s,t)$((ord(t)>1) and (ord(t)<10)) = min(pbacktime(t),
tax*(1.04**(tstep*(t.val-2))) );


solve DICESLR maximizing EU using nlp;
```

References

Bamber, Jonathan. L., and Willy. P. Aspinall. 2013. "An expert judgement assessment of future sea level rise from the ice sheets." *Nature Climate Change*, 3(4): 424–427.

Brooke, Anthony, David Kendrick, and Alexander Meeraus. 1988. "GAMS: A User's Guide. Redwood." *Cal.: The Scientific Press.*

Cai, Yongyang, Kenneth L KL Judd, and Thomas S TS Thomas S TS Lontzek. 2013. "The social cost of stochastic and irreversible climate change." *National Bureau of Economic Research.*

Diaz, Delavane B. 2015. "Integrated Assessment of Climate Catastrophes with Endogenous Uncertainty: Does the Risk of Ice Sheet Collapse Justify Precautionary Mitigation?"

Drud, A S. 1996. "CONOPT: A system for large scale nonlinear optimization." *Reference manual for conopt subroutine library, ARKI Consulting and Development A/S, Bagsvaerd, Denmark.*

Joughin, Ian, Benjamin E Smith, and Brooke Medley. 2014. "Marine ice sheet collapse potentially under way for the Thwaites Glacier Basin, West Antarctica." *Science (New York, N.Y.)*, 344(6185): 735–8.

Kriegler, Elmar, Jim W Hall, Hermann Held, Richard Dawson, and Hans Joachim Schellnhuber. 2009. "Imprecise probability assessment of tipping points in the climate system." *Proceedings of the National Academy of Sciences of the United States of America*, 106(13): 5041–5046.

Lemoine, Derek, and Christian Traeger. 2014. "Watch Your Step: Optimal Policy in a Tipping Climate." *American Economic Journal: Economic Policy*, 6(1): 137–166.

National Research Council. 2013. *Abrupt Impacts of Climate Change: Anticipating Surprises.* Washington, D.C.:The National Academies Press.

Nordhaus, William D. 2010*a*. "DICE-2010 Excel Model."

Nordhaus, William D. 2010*b*. "Projections of Sea Level Rise (SLR)."

Nordhaus, William D., and Paul Sztorc. 2013. "DICE 2013R: Introduction and User's Manual." April.

Olson, Roman, Ryan Sriver, Marlos Goes, Nathan M. Urban, H. Damon Matthews, Murali Haran, and Klaus Keller. 2012. "A climate sensitivity estimate using Bayesian fusion of instrumental observations and an Earth System model." *Journal of Geophysical Research: Atmospheres*, 117: 1–11.

Oppenheimer, Michael. 1998. "Global warming and the stability of the West Antarctic Ice Sheet." *Nature*, 393: 325–332.

Rignot, Eric, J. Mouginot, M. Morlighem, H. Seroussi, and B. Scheuchl. 2014. "Widespread, rapid grounding line retreat of Pine Island, Thwaites, Smith and Kohler glaciers, West Antarctica from 1992 to 2011." *Geophysical Research Letters*, 41(10).

Rutherford, Thomas F. 2013. "A Minimal Framework with Precautionary Mitigation." Snowmass, CO.

Vaughan, David G, and John R Spouge. 2002. "Risk estimation of collapse of the West Antarctic Ice Sheet." *Climatic Change*, 52(1-2): 65–91.